NASA Technical Memorandum 89081

(NASA-TM-89081) DESCRIPTION OF A MAGNETIC BEARING TEST FIXTURE (NASA) 23 p CSCL 14B

N87-20478

Unclas

G3/33 45450

DESCRIPTION OF A MAGNETIC BEARING TEST FIXTURE

NELSON J. GROOM AND WILLIAM L. POOLE

MARCH 1987

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665-5225

SUMMARY

A description of a microcomputer controlled magnetic bearing test fixture is presented. Parameters which are controlled are magnetic bearing current and gaps. Parameters which are measured are magnetic bearing gaps, magnetic flux in the bearing gaps, and bearing forces. The test fixture is configured for bearing elements similar to those used in a laboratory test model Annular Momentum Control Device (AMCD).

INTRODUCTION

This paper describes a microcomputer controlled magnetic bearing test fixture which can be used to develop magnetic bearing actuator control approaches. Parameters which are controlled are magnetic bearing current and gaps. Parameters which are measured are magnetic bearing gaps, magnetic flux in the bearing gaps, and bearing forces. The test fixture is configured for bearing elements similar to those used in a laboratory test model AMCD. The basic AMCD concept is that of a rotating annular rim, suspended by a minimum of three magnetic bearing suspension stations and driven by a noncontacting electromagnetic spin motor. A detailed discussion of the rationale for the AMCD configuration and some of its potential applications is presented in reference 1. A description of the laboratory model is presented in reference 2 and an overview of magnetic bearing linearization and control approaches investigated for the AMCD and other annular magnetically suspended devices is presented in reference 3.

TEST FIXTURE DESCRIPTION

Mechanical

The magnetic bearing test fixture, which is shown in figure 1, uses some of the parts from an earlier model which required manual adjustment of the bearing gaps. For a description of the original test fixture see reference 4. The parts from the original model that have been used include the fixture that holds the magnetic bearing elements, the position sensors and mounts, and the load cell-suspended element assembly. A major change is the way the load cell-suspended element assembly is mounted and positioned. As can be seen in the figure, the load cells are mounted to a bar which in turn is connected, through screw positioners, to geared dc drive motors. The equivalent suspended element can be set to any desired vertical position

in the magnetic bearing gap by controlling the drive motors which turn the screw positioners. The position sensors are used to measure the bearing gaps and provide outputs that are used as feedback signals to control the drive motors. The equivalent suspended element position can be sensed at either end by the position sensors and independently controlled by the drive motors. The magnetic bearing elements shown in the figure have the same dimensions as the original magnetic bearing elements delivered with the laboratory model AMCD (ref. 2). These elements, unlike the original elements, contain no permanent magnet material and use SAE 1010 soft steel as core material. In order to measure magnetic flux in the bearing gaps, a TL3103C linear Hall-effect sensor was mounted in the pole face of each element as shown in figure 2. As can be seen in the figure, the sensors are covered by a brass plate for protection. For more information on the Hall-effect sensors see reference 3.

The description of the load cells presented below is essentially the same as that in reference 5 but is repeated here for completeness. The load cells, shown in figure 1, are strain-gage bridge instrumented bending beams. The output of the bridge is a voltage which is directly proportional to the load applied to the beam. The load cells are connected as shown in the schematic representation of figure 3. They have a load range of ±10 lb(±44.48N) and a nominal scale factor of ±0.2mV/lb-V (±0.45 mV/N-V). Since scale factor and offset differ from cell to cell and vary with changes in test configuration, power supply voltage, and temperature, software was developed to provide periodic system calibration.

Calibration is accomplished by removing the load cell-suspended element assembly and applying a sequence of known loads. A plot of typical calibration data for one cell is presented in figure 4. A first-order, least-squares curve fit is applied to the data to obtain voltage/force coefficients. The test fixture is designed to facilitate easy removal and replacement of the load cell-suspended element assembly.

Measurement and Control

The major system components in the measurement and control system include (1) portable microcomputer system, (2) analog interface, (3) motor power driver, and (4) magnetic bearing element power driver. A schematic representation of the measurement and control system is shown in figure 5. A brief description of the analog interface subsystem is presented below. A more detailed description is presented in Appendix A. Descriptions of the motor power driver and bearing power driver are presented in Appendices B and C, respectively.

The microcomputer used in the system is a Commodore SX-64 [1] which is basically a Commodore 64 microcomputer, 5 inch color monitor, 5 1/4 inch disk drive, and detachable keyboard integrated into a portable configuration. The SX-64 utilizes a 6510 eight bit microprocessor and is programmable in Basic and assembly language. All programs were written in Basic with some of the analog I/O subroutines compiled into machine language and called from the main Basic program.

The analog interface subsystem provides eight multiplexed A/D input channels and one D/A output channel. In the present configuration only 6 input channels are used. These include two sensor, two load cell, and two Hall effect device inputs. The output channel supplies the command for the bearing power drivers. The analog interface system is connected to the user port of the SX-64 and utilizes serial I/O. A more detailed discussion of the analog interface subsystem is presented in Appendix A.

CONCLUDING REMARKS

A description of a microcomputer controlled magnetic bearing test fixture has been presented. The test fixture can provide control of magnetic bearing gap and coil current. Measurements which can be made include magnetic bearing gaps, magnetic flux in the bearing gaps, and bearing forces. The test fixture is configured for bearing elements similar to those used in a laboratory test model AMCD. However, the linearization and control approaches which can be developed using the fixture should be applicable to a wide range of small gap suspension systems.

---

[1] Use of names of manufacturers in this report does not constitute an official endorsement of such manufacturers, either expressed or implied, by the National Aeronautics and Space Administration.

APPENDIX A

ANALOG INTERFACE

A schematic diagram of the analog interface used in the magnetic bearing test fixture is shown in figure A1. The interface consists of an ADC0838 analog to digital (A/D) converter for analog input and an AD558 digital to analog (D/A) converter which provides analog output. The ADC0838 is an 8-bit successive approximation A/D converter with serial I/O and an eight channel multiplexed input. The position sensor, load cell, and Hall cell amplifier outputs are connected to channels 0-5, respectively. The input channels are buffered by LM324 operational amplifiers connected in a voltage follower configuration. Low pass filters are used on the inputs of the voltage followers to reduce noise. For more detailed data on the ADC0383 see reference 5. The AD558 is a voltage-output 8-bit D/A with output amplifier and voltage reference built in. In the configuration shown the D/A output is 0-2.56 V. This output is amplified by a 741 operational amplifier with gain adjusted to provide a 0-3 V input for the magnetic bearing power driver. For more detailed data on the AD558 see reference 6. The analog interface is connected through 74LS126 quad buffers to the user port of the SX-64 microcomputer. The user port provides a complete 8-bit I/O port which is port B of a 6526 Complex Interface Adapter (CIA) chip (see ref. 7). Line 2 (PA2) of port A is also available. The eight I/O lines of port B are labeled PB0-PB7, respectively. PB7 is used to send and receive serial data, PB6 is used to generate the clock signal, and PB4, PB5, and PA2 are used to generate control signals. PB0-PB3 are used to generate control signals for the drive motors and are passed through the analog interface to the motor power amplifiers. A 74LS164 serial in parallel out shift register is used to provide parallel data for the AD558 D/A. The subroutines used to control the ADC0838 and the AD558 are listed in Appendix D.



Figure A1. - Analog interface.

APPENDIX B

MOTOR POWER DRIVER

A schematic diagram of the motor power driver used in the magnetic bearing test fixture is shown in figure B1. The circuit consists of two complementary bridge configurations formed by 2N5680 PNP and 2N5682 NPN power transistors. Each bridge has two inputs which determine the polarity of the voltage applied to the motor. The inputs are driven by two I/O lines from the user port of the SX-64 which are passed through the analog interface. A total of four lines are used and consist of PBO-PB3 as described in Appendix A. The SN7402 quad 2-input NOR gates and SN7404 hex inverters provide logic to turn on the appropriate bridge transistors for a given input and to prevent two transistors on the same side of the bridge from being turned on inadvertently. The motor power amplifier circuit and the inputs from the analog interface are isolated by HCPL2630 optocouplers.



Figure B1. - Motor power driver.

APPENDIX C

MAGNETIC BEARING POWER DRIVER

A schematic diagram of the current driver used in the magnetic bearing test fixture is shown in figure C1. This circuit uses current feedback and is similar in design to the current driver described in reference 4. An input stage with adjustable offset has been added to provide bias current capability. The power driver is capable of supplying up to 5 A to a given magnetic bearing coil and has a current gain of 1 A/V. Drive for the complementary Darlington output stage is provided by an LT1010 buffer amplifier.



Figure C1. - Magnetic bearing power driver.

APPENDIX D

DATA COLLECTION PROGRAM

This appendix presents listings, in Basic, of the main data collection and control program, the analog to digital converter subroutine, and the digital to analog converter subroutine. The subroutines can be appended (with appropriate line number changes) to the main program and used as is. However, in order to minimize the time of conversion, they were compiled to machine language and located in the 4k block of ram starting at address 49152 decimal. The starting address of each routine is given in the listing of the main program. A large number of REMark statements are used in the main program and the subroutines in order to explain program operation.

```
10 REM**************************
20 REM  FLUX FB/64 - SERIAL I/O
30 REM**************************
35 REM-------------------------------------------------------------
40 REM   A = COEFFICIENT IN GAP CONVERSION EQUATION
50 REM   AD = STARTING ADDRESS OF A/D MACHINE LANGUAGE SUBROUTINE
60 REM   A$ = DATE THAT DATA WAS TAKEN
70 REM   B = COEFFICIENT IN GAP CONVERSION EQUATION
80 REM   B$ = TEMPORARY VARIABLE
90 REM   CH = A/D CHANNEL NO. BEFORE CONVERSION
100 REM  CH = VALUE ON GIVEN A/D CHANNEL AFTER CONVERSION
110 REM  C$ = COIL USED TO OBTAIN DATA (UPPER OR LOWER)
120 REM  D = D/A INPUT
130 REM  DA = STARTING ADDRESS OF D/A MACHINE LANGUAGE SUBROUTINE
140 REM  DD = PORT B DATA DIRECTION REGISTER ADDRESS
150 REM  DV = D/A COMMAND INCREMENT
160 REM  D(1) = VALUE OF SENSOR 1 OUTPUT
170 REM  D(2) = VALUE OF SENSOR 2 OUTPUT
180 REM  D(3) = VALUE OF LOAD CELL 1 OUTPUT
190 REM  D(4) = VALUE OF LOAD CELL 2 OUTPUT
200 REM  D(5) = VALUE OF HALL CELL 1 OUTPUT
210 REM  D(6) = VALUE OF HALL CELL 2 OUTPUT
220 REM  D(7) = VALUE OF COIL CURRENT
230 REM  G = GAP IN A/D UNITS
240 REM  G1 = STARTING GAP
250 REM  H1 = HALL CELL 1
260 REM  H2 = HALL CELL 2
270 REM  L1 = LOAD CELL 1
280 REM  L2 = LOAD CELL 2
290 REM  PB = PORT B DATA REGISTER ADDRESS
300 REM  S1 = SENSOR 1
310 REM  S2 = SENSOR 2
320 REM  V = VALUE OF D/A COMMAND
330 REM  V1 = SENSOR 1 FLAG
340 REM  V2 = SENSOR 2 FLAG
350 REM  Z1 = LOAD CELL 1 SCALE FACTOR
360 REM  Z2 = LOAD CELL 2 SCALE FACTOR
365 REM-------------------------------------------------------------
370 REM--LOAD A/D AND D/A MACHINE LANGUAGE SUBROUTINES
380 IFI=0THENI=1:LOAD"ADC0838.ML",8,1
390 IFJ=0THENJ=1:LOAD"AD558.ML",8,1
400 PRINT"♥":R$=CHR$(13):DIM D(7)
410 REM--DEFINE VARIABLES, INITIALIZE PORT B, AND SET D/A OUTPUT TO 0
420 PB=56577:DD=56579:OFF=0:AD=49152:DA=50000:V1=0:V2=0:CH=680:D=686
430 POKEPB,OFF:POKEDD,15:POKED,0:SYSDA
440 REM--INPUT DATA AND OPEN DATA FILE
450 PRINT:PRINT"FLUX FB DATA PROGRAM":PRINT
460 INPUT"DATE=(M/D/Y)";A$
470 INPUT"UPPER COIL(Y/N)";B$
480 IF B$="Y"THEN510
490 OPEN2,8,2,"0:DATA/LC,S,W"
500 C$="LOWER COIL DATA":A=3:B=-1:PRINTC$:GOTO530
510 OPEN2,8,2,"0:DATA/UC,S,W":I1=0:I2=3/255
520 C$="UPPER COIL DATA":A=-1:B=1:PRINTC$
530 PRINT#2,A$;R$;C$
540 INPUT"STARTING GAP=";G1
550 S1=8:S2=12:L1=9:L2=13:H1=10:H2=14
```

```
560 PRINT"LOAD CELL 1 SCALE FACTOR(LB/V)":INPUTZ1
570 PRINT"LOAD CELL 2 SCALE FACTOR(LB/V)":INPUTZ2
580 REM--CONVERT GAP UNITS TO A/D UNITS
590 G=(A*.05+B*G1)*2550
600 REM--LOOP TO SET GAP TO GAP COMMAND VALUE
610 FORI=1TO5
620 REM--COMPARE SENSOR 1 OUTPUT TO GAP COMMAND
630 POKECH,S1:SYSAD:IFPEEK(CH)<GTHEN690
640 IFPEEK(CH)>GTHEN710
650 REM--COMPARE SENSOR 2 OUTPUT TO GAP COMMAND
660 POKECH,S2:SYSAD:IFPEEK(CH)<GTHEN690
670 IFPEEK(CH)>GTHEN710
680 GOTO720
690 GOSUB1010
700 GOTO720
710 GOSUB1220
720 NEXTI
730 REM--INITIALIZE D/A
740 V=0:DV=8
750 REM--INCREMENT D/A OUTPUT
760 POKED,V:SYSDA
770 REM--DELAY TO ALLOW SYSTEM TO SETTLE
780 FORJ=1TO3000:NEXTJ
790 REM--READ A/D CHANNELS AND SCALE RESULTS
800 POKECH,S1:SYSAD:D(1)=PEEK(CH):POKECH,S2:SYSAD:D(2)=PEEK(CH)
810 D(1)=B*((D(1)/2550)-.05*A):D(2)=B*((D(2)/2550)-.05*A)
820 POKECH,L1:SYSAD:D(3)=PEEK(CH):POKECH,L2:SYSAD:D(4)=PEEK(CH)
830 D(3)=5*Z1*D(3)/255:D(4)=5*Z2*D(4)/255
840 POKECH,H1:SYSAD:D(5)=PEEK(CH):POKECH,H2:SYSAD:D(6)=PEEK(CH)
850 D(5)=5*D(5)/255:D(6)=5*D(6)/255
860 D(7)=3*V/255
870 REM--SAVE A/D RESULTS TO DATA FILE
880 FORK=1TO7
890 PRINT#2,D(K)
900 NEXTK
910 REM--INCREMENT D/A COMMAND
920 V=V+DV:IFV>=256THEN950
930 GOTO760
940 REM--INCREMENT GAP COMMAND
950 G1=G1+.005:IFG1>=.150THEN990
960 REM--INITIALIZE D/A
970 POKED,0:SYSDA
980 GOTO590
990 CLOSE2:GOTO1420
1000 REM--SUBROUTINE 1010: SENSOR S1 OR SENSOR S2 OUTPUT LESS THAN GAP COMMAND
1010 POKECH,S1:SYSAD:IFPEEK(CH)<GTHENPOKEPB,1:V1=1
1020 POKECH,S2:SYSAD:IFPEEK(CH)<G AND V1=1THENPOKEPB,5:V2=1
1030 IFPEEK(CH)<G AND V1=0THENPOKEPB,4:V2=1
1040 POKECH,S1:SYSAD:IFPEEK(CH)>=G THEN1070
1050 POKECH,S2:SYSAD:IFPEEK(CH)>=G THEN1140
1060 GOTO1040
1070 POKEPB,OFF:V1=0:V2=0
1080 POKECH,S2:SYSAD:IFPEEK(CH)<GTHEN1100
1090 RETURN
1100 POKEPB,4:V2=1
1110 POKECH,S2:SYSAD:IFPEEK(CH)>=GTHEN1130
1120 GOTO1110
1130 POKEPB,OFF:V2=0:RETURN
1140 POKEPB,OFF:V1=0:V2=0
1150 POKECH,S1:SYSAD:IFPEEK(CH)<GTHEN1170
```

```
1160 RETURN
1170 POKEPB,1:V1=1
1180 POKECH,S1:SYSAD:IFPEEK(CH)>=GTHEN1200
1190 GOTO1180
1200 POKEPB,OFF:V1=0:RETURN
1210 REM--SUBROUTINE 1220: SENSOR S1 OR SENSOR S2 GREATER THAN GAP COMMAND
1220 POKECH,S1:SYSAD:IFPEEK(CH)>GTHENPOKEPB,2:V1=1
1230 POKECH,S2:SYSAD:IFPEEK(CH)>GANDV1=1THENPOKEPB,10:V2=1
1240 IFPEEK(CH)>GANDV1=0THENPOKEPB,8:V2=1
1250 POKECH,S1:SYSAD:IFPEEK(CH)<=G THEN1280
1260 POKECH,S2:SYSAD:IFPEEK(CH)<=G THEN1350
1270 GOTO1250
1280 POKEPB,OFF:V1=0:V2=0
1290 POKECH,S2:SYSAD:IFPEEK(CH)>GTHEN1310
1300 RETURN
1310 POKEPB,8:V2=1
1320 POKECH,S2:SYSAD:IFPEEK(CH)<=GTHEN1340
1330 GOTO1320
1340 POKEPB,OFF:V2=0:RETURN
1350 POKEPB,OFF:V1=0:V2=0
1360 POKECH,S1:SYSAD:IFPEEK(CH)>GTHEN1380
1370 RETURN
1380 POKEPB,2:V1=1
1390 POKECH,S1:SYSAD:IFPEEK(CH)<=GTHEN1410
1400 GOTO1390
1410 POKEPB,OFF:V1=0:RETURN
1420 POKED,0:SYSDA
1430 CLOSE2:END

READY.
```

```
10 REM*******************
20 REM ADC0838 SUBROUTINE
30 REM*******************
40 REM---------------------------------
50 REM   N = INDEX VARIABLE FOR LOOPS
60 REM   A = CHANNEL NUMBER  (ON INPUT)
70 REM   A = VALUE ON GIVEN CHANNEL
80 REM       (ON OUTPUT)
90 REM   D = MUX VARIABLE
100 REM  C = MUX VARIABLE W/ CLOCK PULSE
110 REM  B = DATA BIT
120 REM  M = CONTROL VARIABLE
130 REM--------------------------------
140 REM
150 REM ---- MUX ADDRESS OUTPUT ----
160 REM
170 REM--SET DATA DIRECTION REGISTER FOR OUTPUT
180 POKE 56579,255
190 REM--MASK CONTROL BITS AND PULL CS LOW
200 M=PEEK(56577) AND 15
210 REM--SET START BIT
220 POKE 56577,M OR 128
230 REM--CLOCK PULSE TO SEND START BIT
240 POKE 56577,M OR 192
250 POKE 56577,M OR 128
260 REM--LOOP FOR MUX ADDRESS OUTPUT
270 FOR N=0 TO 3
280 REM--MASK MUX BIT
290 D=A AND 8
300 REM--GET MUX BIT IN MSB POSITION
310 D=D*16
320 REM--PUT MUX BIT IN DATA REGISTER
330 POKE 56577,D OR M
340 REM--SET UP FOR CLOCK PULSE
350 C=DOR64
360 REM--CLOCK PULSE TO SEND MUX BIT
370 POKE 56577,C OR M
380 POKE 56577,D OR M
390 REM--SHIFT A LEFT
400 A=A*2
410 NEXT N
420 REM
430 REM ---- DATA INPUT ----
440 REM
450 REM--USE A FOR DATA BYTE
460 A=0
470 REM--SET DATA DIRECTION REGISTER FOR DATA INPUT
480 POKE 56579,99
490 REM--LOOP FOR DATA INPUT
500 FOR N=0 TO 7
510 REM--CLOCK PULSE
520 POKE 56577,M OR 64
530 POKE 56577,M OR 0
540 REM--READ DATA REGISTER
550 D=PEEK(56577)
560 REM--MASK DATA BIT
570 B=D AND 128
```

```
580 B=B/128
590 REM--SHIFT DATA BIT IN
600 A=A*2+B
610 NEXT N
620 REM--SET CS HIGH
630 POKE 56577,M OR 32
640 RETURN
5000 OPEN4,4:PRINT#4,CHR$(147):CMD4:LIST

READY.
```

APPENDIX D
D/A Subroutine Listing

```
10 REM*****************
20 REM AD558 SUBROUTINE
30 REM*****************
40 REM SERIAL INTERFACE USING SHIFT REGISTER
50 REM------------------------------
60 REM    B=DATA BIT
70 REM    C=PA DATA REGISTER VARIABLE
80 REM    D=D/A VARIABLE
90 REM    N=INDEX VARIABLE FOR LOOPS
100 REM------------------------------
110 REM--SET PB DDR FOR OUTPUT
120 POKE56579,255
130 REM--PRESERVE PA DATA REGISTER CONTENTS AND SET PA2 HIGH
140 C=PEEK(56576)OR4
150 POKE56576,C
160 REM--SET CS (A/D,CS=PB5) AND S1 (SHIFT REGISTER,S1=PB4) HIGH
170 POKE56577,48
180 REM--SHIFT D/A VARIABLE OUT ON PB7
190 FORN=0TO7
200 REM--MASK MSB OF D/A VARIABLE
210 B=D AND 128
220 REM--PUT DATA BIT IN PB REGISTER AND SET CLK LOW WHILE KEEPING CS & S1 HIGH
230 POKE56577,B OR 48
240 REM--LOAD DATA BIT INTO SHIFT REGISTER BY SETTING CLK HIGH
250 POKE56577,B OR 112
260 REM--SHIFT B LEFT
270 B=B*2
280 NEXTN
290 REM--SET PB TO ZERO EXCEPT FOR CS
300 POKE56577,32
310 REM--LATCH DATA INTO D/A BY PULSING PA2 LOW
320 POKE56576,C AND 251
330 POKE56576,C
340 RETURN

READY.
```

REFERENCES

1. Anderson, Willard W.; and Groom, Nelson J.: The Annular Momentum Control Device (AMCD) and Potential Applications. NASA TN D-7866, 1975.

2. Groom, Nelson, J.: Description Of A Laboratory Model Annular Momentum Control Device (AMCD). NASA CP-2346, pp. 157-306, November 1984.

3. Groom, Nelson, J.: Overview Of Magnetic Bearing Control and Linearization Approaches For Annular Magnetically Suspended Devices. NASA CP-2346, pp. 237-306, November 1984.

4. Groom, Nelson J.; and Miller, James B.: A Microprocessor-Based Table Lookup Approach For Magnetic Bearing Linearization. NASA TP-1838, 1981.

5. TL3103I, TL3103C Linear Hall-Effect Sensors. Texas Instruments Inc. May, 1985.

6. ADC0831, ADC0832, ADC0834, and ADC0838 (COP431, COP432, COP434, and COP438) 8-Bit Serial I/O A/D Converters With Multiplexer Options. National Semiconductor Corp., June 1982.

7. Dacport, Low Cost Complete Microprocessor-Compatible 8-Bit DAC, AD558. Analog Devices, Inc., November 1982.

8. Angerhausen, Michael; Becker Achim; Englisch, Lothar; and Gerits, Klaus: The Anatomy of the Commodore 64. Abacus Software, Grand Rapids, MI, c. 1983, 1984.



Figure 1.- Magnetic bearing test fixture.



Figure 2.- Magnetic bearing element.



Figure 3. - Load cell connection diagram.



Figure 4. - Typical calibration data for a given cell.



Figure 5. - Schematic representation of measurement and control system.

Standard Bibliographic Page

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM-89081		
4. Title and Subtitle Description of a Magnetic Bearing Test Fixture		5. Report Date March 1987
		6. Performing Organization Code 505-66-91-01
7. Author(s) Nelson J. Groom and William L. Poole		8. Performing Organization Report No.
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665-5225		10. Work Unit No.
		11. Contract or Grant No.
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546-0001		13. Type of Report and Period Covered Technical Memorandum
		14. Sponsoring Agency Code
15. Supplementary Notes		

16. Abstract

A description of a microcomputer controlled magnetic bearing test fixture is presented. Parameters which are controlled are magnetic bearing current and gaps. Parameters which are measured are magnetic bearing gaps, magnetic flux in the bearing gaps, and bearing forces. The test fixture is configured for bearing elements similar to those used in a laboratory test model Annular Momentum Control Device (AMCD).

17. Key Words (Suggested by Authors(s))	18. Distribution Statement
Control Actuator Magnetic Actuator Magnetic Suspension Magnetic Actuator Test Fixture	Unclassified - Unlimited Subject Category 33

19. Security Classif.(of this report)	20. Security Classif.(of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	23	A02

For sale by the National Technical Information Service, Springfield, Virginia 22161

NASA Langley Form 63 (June 1985)